




PML Library: PD9 - Turnover models with repeated dosing

PD9: Example 1 Objectives

- Characterize the PD of a new anxiolytic compound Zooparc[®] after repeated po dosing
- Data were collected during a Phase II study in psychiatric patients at 5mg and 25 mg dose levels
- Establish turnover characteristics of a biomarker.

PD9: Exploratory Data Analysis

- Observed Response-Time data following repeated QD administration of Zooparc[®] at 5mg and 25mg dose levels



Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

PD9: PK parameters as fixed effects

- The pharmacokinetics of Zooparc[®] have been characterized in a previous study.
- Extravascular 1-compartment model:
- $dC/dt = \text{Dose} \cdot k_a / V \cdot (k_a - k_e)$
 - K_a = absorption rate constant, 1.1 1/hr
 - K_e = elimination rate constant, 0.128 1/hr
 - V = volume of distribution, 5 L/kg
- Solve concentration function at different time points for a given dose level.

PD9: Turnover model: inhibition of k_{in}

- Rate of production is inhibited by the function $I(C)$, assuming $I_{max} = 1$
- Animal studies have shown that complete inhibition of response is possible, i.e. $I(C) = 0$ at high concentrations
- Indirect response is modeled using $dR/dt = k_{in} \cdot I(C) - k_{out} \cdot R$
 - K_{in} = zero order rate of production
 - K_{out} = first order rate of loss
 - $I(C)$ = mechanistic inhibition of K_{in}

PD9: Objective: Model drug mechanism using Inhibitory function

- $I(C)$ – typical inhibitory I_{max} function
- Inhibitory function $I(C) = 1 - I_{max} * C^\gamma / (IC_{50}^\gamma + C^\gamma)$
 - I_{max} = maximum inhibitory response
 - IC_{50} = concentration at which 50% of maximum inhibition occurs
 - γ = exponent for sigmoid I_{max} model

PD9: Objective: Full mechanistic Turnover model

- Full model, substituting in the mechanistic inhibition function:
- $$dR/dt = K_{in} * (1 - I_{max} * C^{\gamma} / (C^{\gamma} + IC_{50}^{\gamma})) - K_{out} * R$$
- Get final estimates for the following parameters:
 - K_{in} = zero order rate of production
 - K_{out} = 1st order loss
 - I_{max} = maximum inhibition
 - IC_{50} = concentration at which 50% of maximum inhibition occurs
 - γ = exponent for sigmoid I_{max} function

PD9: Built-in PK/indirect: inhibition of production

☐ Population? | **Structure** | Parameters | Input Options | Initial Estimates | Run Options | Model Text | Plots | no warnings

Type: **PK/Indirect** Set WNL Model

Parameterization: **Micro** Absorption: **Extravascular** Num Compartments: **1**

☐ tlag? ☐ Ka = Ke ? ☐ Elim. Cpt. ?


☐ Closed form? ☒ **Freeze PK?**

Indirect: **Inhib. Limited** OF **Build-up**
Exponent ☐ Freeze Indirect?

Residual Error:
E **EObs** **EEps** = **Additive** ☐ BQL?
Stdev: **1** 0.81236 Accept
☐ Freeze

Parameters:	Statements:
Ka	deriv(Aa = - Ka * Aa)
V	deriv(A1 = Ka * Aa - Ke * A1)
Ke	deriv(E = Kin * (1 - Imax * C ^ gam / (C ^ gam + IC50 ^ gam)) -
Kin	dosepoint(Aa)
Kout	C = A1 / V
Imax	sequence{E = Kin / Kout}
IC50	error(EEps = 1)
gam	observe(EObs = E + EEps)

PD9: Graphical Model



Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

PD9: Indirect Response Model: PML Code

```
1 test(){
2   # differential equations for PK model
3   deriv(Aa = - Ka * Aa)
4   deriv(A1 = Ka * Aa - Ke * A1)
5   # differential equation for Indirect Response model
6   deriv(E = Kin * (1 - Imax * C ^ gam / (C ^ gam + IC50 ^ gam)) - Kout * E)
7   # extravascular dose administration
8   dosepoint(Aa)
9   # concentration in the central compartment
10  C = A1 / V
11  # baseline response R0
12  sequence{E = Kin / Kout}
13  # observed response and error model
14  error(EEps = 1)
15  observe(EObs = E + EEps)
16  # PK parameters as frozen fixed effects
17  fixef(Ka(freeze) = c(, 1.1, ))
18  fixef(V(freeze) = c(, 5, ))
19  fixef(Ke(freeze) = c(, 0.128, ))
20  # Indirect response parameters with initial estimates
21  fixef(Kin = c(, 4.8, ))
22  fixef(Kout = c(, 0.06, ))
23  fixef(Imax(freeze) = c(, 1, ))
24  fixef(IC50 = c(, 0.25, ))
25  fixef(gam = c(, 2, ))
26 }
```

Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

PD9: Initial Estimates



- $R_0 = 80$, from exploratory plot response at time = 0
- Onset slope = $-k_{out} = -(\ln(R_1) - \ln(R_2)) / (t_1 - t_2)$
- $-K_{out} = -\ln(80/48) / (0-8) = -0.06 \text{ hr}^{-1}$
- $K_{in} = R_0 \cdot k_{out} = 80 \cdot 0.06 = 4.8 \text{ hr}^{-1}$
- $I_{max} = 1$
- IC_{50} is obtained by first finding the time of 50% inhibition, then solving the PK concentration function at that same time point. $IC_{50} = 0.25 \text{ mg/L}$
- Gamma (exponent) = 2, obtained from Initial Estimates tab



Example 1 Demo

PD9: Example 1 Results

- Fit



- Final Parameter Estimates

Parameter	Estimate	Units	Stderr	CV%	2.5% CI	97.5% CI	Var. Inf. factor
Ka	1.1	1/hr	0	0	1.1	1.1	0
V	5		0	0	5	5	0
Ke	0.128	1/hr	0	0	0.128	0.128	0
Kin	8.80231		0.31140006	3.5377084	8.1680101	9.4366099	0.059228
Kout	0.105859	1/hr	0.0028807291	2.7212888	0.099991159	0.11172684	4.6806E-06
Imax	1		0	0	1	1	0
IC50	0.244842		0.0053673085	2.1921519	0.23390917	0.25577483	3.2596E-05
gam	1.38354		0.036746428	2.6559715	1.3086901	1.4583899	0.001387
stdev0	2.57322		0.0088998876	0.34586579	2.5550916	2.5913484	


Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)



Example 2 Model 1: Inhibition of Production

Example 2 Exploratory Data Analysis

- Observed Response-Time data following two rapid infusions



Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

Example 2 Objectives

- Mechanism of action is not known. Develop two different Turnover models for Response-Time data.
- Model 1: inhibition of production
 - $dR/dt = k_{in} * I(C) - k_{out} * R$
- Model 2: stimulation of loss
 - $dR/dt = k_{in} - k_{out} * S(C) * R$
- Compare the models using parameter precision and diagnostics (AIC, BIC, -2LL)

Example 2 PK parameters as fixed effects

- The PK of compound 2 has been characterized in a previous study.
- IV infusion 3-compartment model with clearance parameters:
 - $dA1/dt = -Cl * C - Cl2 * (C - C2) - Cl3 * (C - C3)$
 - $dA2/dt = Cl2 * (C - C2)$
 - $dA3/dt = Cl3 * (C - C3)$
 - $C = A1 / V$
 - $C2 = A2 / V2$
 - $C3 = A3 / V3$
- PK parameter estimates:
 - $V = 0.7633$
 - $V2 = 1.72876$
 - $V3 = 3.43857$
 - $Cl = 6.2417$
 - $Cl2 = 5.4595$
 - $Cl3 = 0.85806$

Ex2: Model 1 drug mechanism using Inhibitory function

- $I(C)$ – typical inhibitory I_{max} function
- Inhibitory function $I(C) = 1 - I_{max} * C^{\gamma} / (IC_{50}^{\gamma} + C^{\gamma})$
 - I_{max} = maximum inhibitory response
 - IC_{50} = concentration at which 50% of maximum inhibition occurs
 - γ = exponent for sigmoid I_{max} model

Ex2: Model 1 Full mechanistic Turnover model

- Full model, substituting in the mechanistic inhibition function:
- $$dR/dt = K_{in} * (1 - I_{max} * C^{\gamma} / (C^{\gamma} + IC_{50}^{\gamma})) - K_{out} * R$$
- Get final estimates for the following parameters:
 - K_{in} = zero order rate of production
 - K_{out} = 1st order loss
 - I_{max} = maximum inhibition
 - IC_{50} = concentration at which 50% of maximum inhibition occurs
 - γ = exponent for sigmoid I_{max} function

Ex2: Built-in PK/indirect: inhibition of production

☐ Population? | Structure | Parameters | Input Options | Initial Estimates | Run Options | Model Text | Plots | no warnings

Type: PK/Indirect Set WNL Model

Parameterization: Micro | Absorption: Extravascular | Num Compartments: 1

☐ tlag? ☐ Ka = Ke ? ☐ Elim. Cpt. ?

☐ Closed form? ☒ Freeze PK?


☐ Infusions possible? ☐ Effect Cpt?

Indirect: Inhib. Limited OF Build-up
Exponent ☐ Freeze Indirect?

Residual Error:
E EObs EEps = Additive ☐ BQL?
Stdev: 1 0.81236 Accept
☐ Freeze

Parameters:	Statements:
Ka	deriv(Aa = - Ka * Aa)
V	deriv(A1 = Ka * Aa - Ke * A1)
Ke	deriv(E = Kin * (1 - Imax * C ^ gam / (C ^ gam + IC50 ^ gam)) -
Kin	dosepoint(Aa)
Kout	C = A1 / V
Imax	sequence{E = Kin / Kout}
IC50	error(EEps = 1)
gam	observe(EObs = E + EEps)

Ex2: Model 1 Graphical



Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

Ex2: Model 1 inhibition of production: PML Code

```
1 test(){
2   # differential equations for the PK model
3   deriv(A1 = - C1 * C - C12 * (C - C2) - C13 * (C - C3))
4   deriv(A2 = C12 * (C - C2))
5   deriv(A3 = C13 * (C - C3))
6   # differential equation for Indirect Response model
7   deriv(E = Kin * (1 - Imax * C ^ gamma / (C ^ gamma + IC50 ^ gamma)) - Kout * E)
8   # IV dose administration
9   dosepoint(A1)
10  # concentrations in central and peripheral compartments
11  C = A1 / V
12  C2 = A2 / V2
13  C3 = A3 / V3
14  # baseline response R0
15  sequence(E = Kin / Kout)
16  # observed response and error model
17  error(Eeps = 1)
18  observe(EObs = E + Eeps)
19  # PK parameters as frozen fixed effects
20  fixef(V(freeze) = c(, 0.7633, ))
21  fixef(V2(freeze) = c(, 1.72876, ))
22  fixef(V3(freeze) = c(, 3.43857, ))
23  fixef(C1(freeze) = c(, 6.2417, ))
24  fixef(C12(freeze) = c(, 5.4595, ))
25  fixef(C13(freeze) = c(, 0.85806, ))
26  # Indirect response parameters with initial estimates
27  fixef(Kin = c(, 48.4, ))
28  fixef(Kout = c(, 4.4, ))
29  fixef(Imax = c(, 1, ))
30  fixef(IC50 = c(, 300, ))
31  fixef(gamma = c(, 4, ))
32 }
```

Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

Ex2 Model 1: Initial Estimates

- $R_0 = 11$, from exploratory plot response at time = 0
- Onset slope = $-k_{out} = -(\ln(R_1) - \ln(R_2)) / (t_1 - t_2)$
- $-K_{out} = -\ln(7.5/2.5) / (0-0.25) = -4.4 \text{ hr}^{-1}$
- $K_{in} = R_0 \cdot k_{out} = 11 \cdot 4.4 = 48.4 \text{ hr}^{-1}$
- $I_{max} = 1 - R_{min}/R_0 = 1 - 2/11 = 0.82$
- IC_{50} is obtained by first finding the time of 50% inhibition, then solving the PK concentration function at that same time point. $IC_{50} = 250 \text{ mg/L}$
- Gamma (exponent) = 10, obtained from Initial Estimates tab

Ex2 Run Options: add a table with more predicted points

☐ Population?

General | Parameters | Input Options | Initial Estimates | **Run Options** | Model Text | Plots | no warnings

Method: Naive-pooled
N Iter: 1000
☒ Sort Input?
Max ODE: matrix expone ▾

Stderr: Central Dil ▾
Confidence Level %: 95
Partial Deriv dP: 0.00001
P. D. Time Steps: 20
Advanced >>

Run Mode
☒ Simple
☐ Simulation

Table01 ☒
Add Table
Structural Parameter
Times: seq(0,2.5,0.1)
When covr set:
When dose:
When observe:
Variables: C,E
☐ TAD ☐ IRES ☐ Weight ☐ IWRES



Example 2 Model 2: Stimulation of Loss

Ex2: Model 2 drug mechanism using Stimulatory function

- $S(C)$ – typical stimulatory Emax function
- Stimulatory function $S(C) = (1 + E_{\max} * C^{\gamma} / (C^{\gamma} + EC50^{\gamma}))$
 - E_{\max} = maximum stimulatory response
 - $EC50$ = concentration at which 50% of maximum stimulation occurs
 - γ = exponent for sigmoid Imax model

Ex2: Model 2 Full mechanistic Turnover model

- Full model, substituting in the mechanistic inhibition function:
- $$dR/dt = K_{in} - K_{out} * (1 + E_{max} * C^{\gamma} / (C^{\gamma} + EC50^{\gamma})) * R$$
- Get final estimates for the following parameters:
 - K_{in} = zero order rate of production
 - K_{out} = 1st order loss
 - E_{max} = maximum stimulation
 - $EC50$ = concentration at which 50% of maximum stimulation occurs
 - γ = exponent for sigmoid E_{max} function

Ex2: Built-in PK/indirect: stimulation of loss

☐ Population?

StructureParametersInput OptionsInitial EstimatesRun OptionsModel TextPlotsno warnings

Type: PK/Indirect

Set WNL Model

Parameterization:Clearance

Absorption:Intravenous

Num Compartments:3

☐ Saturating?☐ tlag?☐ Elim. Cpt?

☐ Closed form?

☒ Infusions possible?☒ Freeze PK?

☐ Effect

Indirect:

Stim. Limited

OF

Loss

Exponent

☐ Freeze Indirect?

Residual Error:

E

EObs

EEps

=

Additive

☐ BQL?

Stdev:

1

0.568381

Accept

☐ Freeze

Parameters:

Statements:

V

V2

V3

Cl

Cl2

Cl3

Kin

Kout

E_{max}

EC₅₀

gam

deriv(A1 = - Cl * C - Cl2 * (C - C2) - Cl3 * (C - C3))

deriv(A2 = Cl2 * (C - C2))

deriv(A3 = Cl3 * (C - C3))

deriv(E = Kin - Kout * (1 + E_{max} * C^{gam} / (C^{gam} +

dosepoint(A1)

C = A1 / V

C2 = A2 / V2


C3 = A3 / V3

sequence{E = Kin / Kout}

error(EEps = 1)

observe(EObs = E + EEps)

Ex2: Model 2 Graphical



Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

Ex2: Model 2 stimulation of loss: PML Code

```
1 test(){
2     # differential equations for the PK model
3     deriv(A1 = - C1 * C - C12 * (C - C2) - C13 * (C - C3))
4     deriv(A2 = C12 * (C - C2))
5     deriv(A3 = C13 * (C - C3))
6     # differential equation for Indirect Response model
7     deriv(E = Kin - Kout * (1 + Emax * C^gamma / (C^gamma + EC50^gamma)) * E)
8     # IV dose administration to central compartment
9     dosepoint(A1)
10    # concentrations in central and peripheral compartments
11    C = A1 / V
12    C2 = A2 / V2
13    C3 = A3 / V3
14    # baseline response R0
15    sequence{E = Kin / Kout}
16    # observed response with error model
17    error(Eeps = 1)
18    observe(EObs = E + Eeps)
19    # PK parameters as frozen fixed effects
20    fixef(V(freeze) = c(, 0.7633, ))
21    fixef(V2(freeze) = c(, 1.72876, ))
22    fixef(V3(freeze) = c(, 3.43857, ))
23    fixef(C1(freeze) = c(, 6.2417, ))
24    fixef(C12(freeze) = c(, 5.4595, ))
25    fixef(C13(freeze) = c(, 0.85806, ))
26    # Indirect response parameters with initial estimates
27    fixef(Kin = c(, 48.4, ))
28    fixef(Kout = c(, 4.4, ))
29    fixef(Emax = c(, 4.5, ))
30    fixef(EC50 = c(, 350, ))
31    fixef(gamma = c(, 1, ))
32 }
```

Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

Ex2 Model 2: Initial Estimates

- $R_0 = 11$, from exploratory plot response at time = 0
- Onset slope = $-k_{out} = - (\ln(R_1) - \ln(R_2)) / (t_1 - t_2)$
- $-K_{out} = - \ln(7.5/2.5) / (0-0.25) = -4.4 \text{ hr}^{-1}$
- $K_{in} = R_0 \cdot k_{out} = 11 \cdot 4.4 = 48.4 \text{ hr}^{-1}$
- $E_{max} = R_0 / R_{min} - 1 = 11/2 - 1 = 4.5$
- EC50 is obtained by first finding the time of 50% recovery, then solving the PK concentration function at that same time point. $EC_{50} = 350 \text{ mg/L}$
- Gamma (exponent) = 10, obtained from Initial Estimates tab



Example 2 Demo

Example 2 Results comparison

- Overall

	Source	RetCode	-2LL	AIC	BIC	nParm
1	Model inh Kin	1	44.87176	56.87176	66.537267	6
2	Model stim Kout	1	63.19406	75.19406	84.859567	6


- Final Parameter Estimates

	Source	Parameter	Estimate	CV%
7	Model stim Kout	EC50	309.902	5.4251966
8	Model stim Kout	Emax	4.38766	8.6242905
9	Model inh Kin	gamma	7.0629	12.506123
10	Model stim Kout	gamma	21.533	88.677962
11	Model inh Kin	IC50	244.203	2.2904303
12	Model inh Kin	Imax	0.830643	1.423949
13	Model inh Kin	Kin	143.314	8.7613725
14	Model stim Kout	Kin	34.3159	10.159728
15	Model inh Kin	Kout	12.5528	7.2383399
16	Model stim Kout	Kout	2.87789	10.617335
17	Model inh Kin	stdev0	0.443719	11.6247
18	Model stim Kout	stdev0	0.56838	11.624738

Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

Example 2: Overlay Plot

- Overlay predicted responses for both models with observed data:



Questions?



PML School: Materials

- Each model will be made available in Certara Forum
 - Link to live webinar and presentation slides
 - <https://support.certara.com/forums/forum/34-pml-school/>
 - Model text as file download
 - Can be imported into Phoenix model object to be run on a new dataset
 - Questions and comments can be exchanged in the Forum
 - Or can be entered into the Certara Support portal at:
 - <https://support.certara.com/support>
 - Or can be sent as emails to support@certara.com

PML School: Channels

- Notifications on updates of materials, Q&A and discussions
- Announcements of new sessions
 - PML School on LinkedIn:
 - If you have a LinkedIn Account, please join our new LinkedIn Group:
 - <https://www.linkedin.com/groups/8614342>
 - PML School on Twitter: @PML_School
 - PML School on Youtube:
 - <https://www.youtube.com/user/CertaraLP/videos>

- A wide range of On Demand and Classroom courses are available through Certara University:
 - Introductory, intermediate and advanced instruction in Phoenix WinNonlin, Population Modeling using NLME, IVIVC Toolkit
 - Fundamentals of Pharmacokinetics and Pharmacodynamics
 - Noncompartmental data analysis
 - Programming Bootcamp
 - Partner Lectures and Webinar series
- Please visit our [Certara University](#) web site for more information

Coming up...



Dose-response-time Analysis I-IV

Analyze dose-response-time data with an instantaneous effect model

July 27, 2017 | 10am EST

Presenter: Bernd Wendt